# SLOWING DOWN IN HYDROGEN OR PARTICLE TRANSMISSION THROUGH A SLAB SHIELD

© M. Ragheb
11/3/2013

## INTRODUCTION

We consider the problem of neutrons slowing down in hydrogen with macroscopic scattering and absorption cross sections:

$$\sum_s \ , \sum_a$$

assumed constant, the source being at zero lethargy and monoenergetic. The objective of the calculation is to determine the slowing down density $\psi(x)$, a s a function of x. The resonance escape probability is the slowing down density per unit source, at the thermal lethargy x = T.

The energy dependent problem described above has an equivalent in the spatial variable. In this case it is equivalent to that of a mono-energetic beam of particles (neutrons or gamma rays), normally incident on an infinite slab shield with constant cross sections:

$$\sum_s \ , \sum_a \ , and \sum_t \ ,$$

but the result of each collision event being absorption or un-deflected straight-ahead progress into the shield. The objective of the calculation is the determination of the spatial attenuation of the impinging beam $\psi(x)$. The shield transmission probability through a thickness x = T mean free paths will be $\psi(T)$.

Both of these problems are represented by the following integral equation for $\psi(x)$, which is a Volterra equation of the second type:

$$\psi(x) = S(x) + \int_0^x K(x,y)\psi(y)dy, \quad 0 \le x < \infty, \qquad (1)$$

where x extends over the positive real line. The kernel is interpreted as:

"If y then x with probability density K(x, y)."

## MATHEMATICAL FORMULATION

The problem can be described mathematically as:

$$S(x) = e^{-x} \qquad 0 < x < \infty$$

$$K(x,y) = \begin{cases} 0 & 0 < y \le x \le \infty \\ \dfrac{\sum_s (x)}{\sum_t (x)} e^{-(x-y)} & 0 < y \le x \le \infty \end{cases} \tag{2}$$

where: S(x) is a first event source,

K(x, y) is a kernel that does not allow backward steps,

$\dfrac{\sum_s (x)}{\sum_t (x)}$ is the survival probability,

$\dfrac{\sum_a (x)}{\sum_t (x)}$ is the absorption probability.

The latter probabilities satisfy the relationships:

$$0 < \frac{\sum_s (x)}{\sum_t (x)} = 1 - \frac{\sum_a (x)}{\sum_t (x)}, \qquad 0 < x < \infty. \tag{3}$$

In the present case the transport kernel K(x, y) is factorable into a collision kernel C(x) and a transport kernels T(x) as:

$$\text{Collision kernel:} \qquad C(x) = \frac{\sum_s (x)}{\sum_t (x)} \tag{4}$$

$$\text{Transport kernel:} \qquad T(x) = e^{-(x-y)} \tag{5}$$

## ANALYTICAL SOLUTION FOR THE NEUMANN SERIES TERMS

For comparison to the Monte Carlo results later, we can find an analytical solution to the problem at hand as a Neumann series solution which can be written as:

$$\psi(x) = S(x) + \sum_{i=1}^{\infty} I_n(x) \tag{6}$$

where:

$$I_n(x) = \int ... \int L_n(x; y_1, y_2, \ ... \ , y_n) dy_1 dy_2 \ ... \ dy_n \tag{7}$$

$$L_n(x; y_1, y_2, y_3, \ ... \ , y_n) = K(x, y_n) K(y_n, y_{n-1}) \ ... \ K(y_2, y_1) S(y_1) \text{ , n = 1, 2, 3, …} \tag{8}$$

substituting for S(x) and K(x, y):

$$\psi(x) = e^{-x} + \int_0^L \psi(y) \frac{\sum_s(x)}{\sum_t(x)} e^{-(x-y)} dy, \;\; 0 < x < \infty \qquad (9)$$

Thus:

$$\begin{aligned} L_n(x) &= \left(\frac{\Sigma_s}{\Sigma_t}\right)^n e^{-(x-y_n)} e^{-(y_n - y_{n-1})} \;\; \ldots \;\; e^{-(y_2 - y_1)} . e^{-y_1} \\ &= \left(\frac{\Sigma_s}{\Sigma_t}\right)^n . e^{-x} \end{aligned} \qquad (10)$$

We can estimate the Neumann series terms:

$$I_0 = e^{-x}$$

$$I_1 = \left(\frac{\Sigma_s}{\Sigma_t}\right)^1 e^{-x} \int_0^x dy_1$$

$$I_2 = \left(\frac{\Sigma_s}{\Sigma_t}\right)^2 e^{-x} \int_0^x \int_0^{y_1} dy_1 dy_2$$

$$\ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots$$

$$I_n = \left(\frac{\Sigma_s}{\Sigma_t}\right)^n e^{-x} \int_0^x \int_0^{y_1} \int_0^{y_2} \ldots\ldots\ldots\ldots \int_0^{y_{n-1}} dy_1 dy_2 dy_3 \;\ldots\; dy_n$$

By direct integration we get:

$$I_0 = e^{-x}$$

$$I_1 = \left(\frac{\Sigma_s}{\Sigma_t}\right)^1 e^{-x} \frac{x}{1}$$

$$I_2 = \left(\frac{\Sigma_s}{\Sigma_t}\right)^2 e^{-x} \frac{x^2}{2.1}$$

$$\ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots$$

$$I_n = \left(\frac{\Sigma_s}{\Sigma_t}\right)^n e^{-x} \frac{x^n}{n!}$$

The following recurrence relation is satisfied:

$$I_{n+1} = I_n \left(\frac{\Sigma_s}{\Sigma_t}\right) e^{-x} \frac{x}{(n+1)} \tag{11}$$

Summing up the series, we obtain:

$$\begin{aligned}
\psi(x) &= \sum_{i=1}^{\infty} I_i \\
&= e^{-x}[1 + \left(\frac{\Sigma_s}{\Sigma_t}\right)^1 \frac{x}{1!} + \left(\frac{\Sigma_s}{\Sigma_t}\right)^2 \frac{x^2}{2!} + \left(\frac{\Sigma_s}{\Sigma_t}\right)^3 \frac{x^3}{3!} + \ldots + \left(\frac{\Sigma_s}{\Sigma_t}\right)^n \frac{x^n}{n!} \\
&= e^{-x} e^{+\frac{\Sigma_s}{\Sigma_t} x} \\
&= e^{-(1-\frac{\Sigma_s}{\Sigma_t})x} \\
&= e^{-\frac{\Sigma_a}{\Sigma_t} x}, \quad 0 < x < \infty
\end{aligned} \tag{12}$$

Note here that x is measured in units of mean free paths. This analytical solution to the problem can be used as a benchmark for comparison to a result that can be generated through a Monte Carlo simulation.

## MONTE CARLO SOLUTION

To simulate the particle transport process in the above problem we need to devise ways of sampling the transport kernel and the collision kernel.

As the probability density function for our generated chain we choose:

$$p(x,y) = \frac{T(x,y)}{\int_y^{\infty} T(x,y)dx} = \frac{e^{-(x-y)}}{\int_y^{\infty} e^{-(x-y)}dx} = \frac{e^{-(x-y)}}{e^{y}.\left[e^{-x}\right]_{\infty}^{y}} = \frac{e^{-(x-y)}}{1} = e^{-\Delta d} \quad (13)$$

where: $\Delta d = (x - y),$ and the transport kernel is properly normalized.

The cumulative distribution function for p(x, y) is:

$$C(x,y) = \frac{\int_y^{x} T(x',y)dx'}{\int_y^{\infty} T(x,y)dx} = \frac{\int_y^{x} e^{-(x'-y)}dx'}{1} = \left[e^{-(x'-y)}\right]_{x}^{y} = 1 - e^{-(x-y)} = 1 - e^{-\Delta d} = \rho \quad (14)$$

Generating a uniform random number over the unit interval enables us to sample the positions $x_i$ of the colliding particle over the interval $\left[0,\infty\right]$ as:

$$x_i = x_{i-1} + \Delta d_i \quad (15)$$

where: $$\Delta d_i = -\ln(1-\rho_i) = -\ln \rho_i \ , \ i = 1,2,3, \ldots , n. \quad (16)$$

Here, we replace $(1-\rho_i)$ by $\rho_i$ since they are both uniformly distributed over the unit interval.

It is worth noting that we choose the variable $\Delta d$ since it is the incremental distance to the next collision that is customarily sampled in particle transport simulations. The procedure shown in Fig. 1 simulates the described problem and generates the Monte Carlo solution for comparison to the exact analytical solution. Statistical weighting is carried out by adjusting the statistical weight of the particle by its survival probability $\frac{\Sigma_s}{\Sigma_t}$ at each collision in lieu of allowing absorptions.

```
!       Transport1 f90
!       Particle Transport Problem
!       Slowing Down in Hydrogen, or Particle Transmission Through a slab
!       shield with asymmetric collisions
!       M. Ragheb
        dimension x(21),y(21),xm(2),xxb(20),yy(20)
!       Initialize variables
!  Minimum and Maximum values on x axis
        data xm/0.0,20./
!       c=absorption probability
!       cc=scattering probability
        data c/0.1/
!       nrg=number of spatial regions or energy groups
        data nrg/20/
!       nmfp=number of mean free paths
        data nmfp/20/
!       n=number of generated chains or histories
        data n/10000/
!       nc=maximum chain length
        data nc/30/
!       Analytical solution
        f(xx)=exp(-xx*c)
        cc=1.0-c
        xn=n
        write(*,*) n
!  Open output file
        open(44, file='random_out')
!       Generate exact analytical solution
        x(1)=0.0
        do i=1,nmfp
                x(i+1)=x(i)+1.0
        end do
!       Calculate average value over each interval
        do i=1,nmfp
                y(i)=(f(x(i))-f(x(i+1)))/c
        end do
        write(*,*) x
        write(*,*) y
!       Determine regions boundaries
        dx=(xm(2)-xm(1))/nrg
        xxb(1)=dx
        do i=2,nrg
                xxb(i)=xxb(i-1)+dx
        end do
!       Generate particle histories
        do i=1,n
!               Sample source
                call random(rr)
                xxx=-log(rr)
                weight=1.0
!               Score contribution of source term
```

```
                    if(xxx.LE.xxb(1)) then
                            yy(1)=yy(1)+weight
                    end if
                    do j=2,nrg
                        if((xxx.LE.xxb(j)).AND.(xxx.GT.xxb(j-1))) then
                            yy(j)=yy(j)+weight
                        end if
                    end do
!               Generate rest of collision chain
                do k=1,nc
!                   Transport particle to new position
                    call random(rr)
                    xxx=xxx-log(rr)
!                   Apply collision kernel
                    weight=weight*cc
!                   Score collision terms in appropriate bins
                    if(xxx.LE.xxb(1)) then
                            yy(1)=yy(1)+weight
                    end if
                    do l=2,nrg
                        if((xxx.LE.xxb(l)).AND.(xxx.GT.xxb(l-1))) then
                            yy(l)=yy(l)+weight
                        end if
                    end do
                end do
            end do
!           Final scores
            do i=1,nrg
                yy(i)=yy(i)/(xn*dx)
                if(yy(i).LE.1.0E-20) then
                    yy(i)=1.0E-20
                end if
            end do
            write(*,*)xxb
            write(*,*)yy

!           Write results on output file
            do i=1,nmfp
                write(44,*) xxb(i), yy(i), y(i)
            end do
            end
```

Figure 1. Monte Carlo procedure for slowing down in hydrogen or particle transmission through a slab shield with asymmetric collisions.

A comparison between the Monte Carlo simulation results and the exact analytical result for different numbers of histories, N=10,100, 1,000 and 10,000 are shown in Figs. 2 to 5. The case pertains to a highly scattering medium with a survival probability of 0.9. The number of Neumann series terms considered is 30 terms.




Figure 2. Comparison of Monte Carlo result to exact analytical result, N=10.




Figure 3. Comparison of Monte Carlo result to exact analytical result, N=100.




Figure 4. Comparison of Monte Carlo result to exact analytical result, N=1000.




Figure 5. Comparison of Monte Carlo result to exact analytical result, N=10,000.

**BIASING THE TRANSPORT KERNEL: THE EXPONENTIAL TRANSFORMATION METHOD**

The probability density function for inter-collision distances in a an infinite medium is:

$$P_d(x) = \Sigma_t e^{-\int_0^x \Sigma_t(s)ds} \tag{17}$$

where x is the distance from the preceding collision along the direction of motion of the particle.

For an infinite homogeneous medium it specializes to:

$$P_d(x) = \Sigma_t e^{-\Sigma_t x} \;\; , \;\; 0 < x < \infty \tag{18}$$

This equation is normally used for sampling flight lengths in particle transport simulations, even when finite media are used, but escape is allowed to restore normalization.

A sampled flight can be obtained from the cumulative distribution function:

$$\begin{aligned} C(x) &= \int_0^x \Sigma_t e^{-\Sigma_t x} dx \\ &= +\frac{\Sigma_t}{\Sigma_t}[e^{-\Sigma_t x}]_x^0 \\ &= 1 - e^{-\Sigma_t x} = \rho \end{aligned} \tag{19}$$

where $\rho$ is a uniformly distributed random number over the unit interval.

Inverting the cumulative distribution function for x yields;

$$\begin{aligned} 1 - \rho &= e^{-\Sigma_t x} \\ x &= -\frac{1}{\Sigma_t}\ln(1-\rho) = -\frac{1}{\Sigma_t}\ln \rho \end{aligned} \tag{20}$$

The exponential transformation method as an importance sampling method considers the modified probability density function for the transport kernel:

$$P_d[x, \beta(\mu)] = \beta(\mu)\Sigma_t e^{-\beta(\mu)\Sigma_t x} \;\; , \;\; 0 < x < \infty \tag{21}$$

where: $\mu = \cos\theta$ is the cosine of the scattering angle $\theta$,

$\beta(\mu) = 1 - \mu c$ is the biasing parameter arising from the exponential transformation with absorption weighting,

$c = 1 - \alpha$ is an adjustable constant,
$\alpha$ is a parameter which is also adjustable,
$\beta(1) = 1 - c = 1 - 1 + \alpha = \alpha,$ for right moving particles,
$\beta(-1) = 1 + c = 2 - \alpha,$ for left moving particles.

The application of the exponential transformation means that the total cross section is modified from:

$$\Sigma_t \ to \ \beta(\mu).\Sigma_t \, ,$$

so the mean free path traveled by the particle changes from:

$$\frac{1}{\Sigma_t} \ to \ \frac{1}{\beta(\mu)}.\frac{1}{\Sigma_t} \, .$$

Now $\beta$ depends on the particle direction, as one would like to stretch particle tracks to the right to penetrate a shield with $\beta$ < 1, and shrink them to the left with $\beta$ > 1. Notice that the case with $\beta$ = 1 corresponds to the analog case without application of track stretching or shrinking within the exponential transformation.

The probability density function $P_d$ can be considered as the product of two factors: the probability:

$$e^{-\beta \Sigma_t x}$$

of traveling x mean free paths without collision and the probability per unit length:

$$\beta \Sigma_t$$

of colliding at x. To correctly restore the expected value of the weight density of the particles, one multiplies the particles statistical weight by the ratio of the analog to the non-analog probabilities:

$$\begin{aligned} n_R &= \frac{\Sigma_t e^{-\Sigma_t d_n}}{\beta \Sigma_t e^{-\beta \Sigma_t d_n}} \\ &= \frac{1}{\beta} e^{-d_n (1-\beta) \Sigma_t} \end{aligned} \tag{22}$$

where: $d_n$ is the distance traveled at the n-th collision.

The particle's statistical weight at the n-the collision will be:

$$W_n = W_{n-1}.P_s.n_R$$
$$= W_{n-1}.P_s.\frac{1}{\beta} e^{-d_n(1-\beta)\Sigma_t}, \quad n = 1, 2, \ ... \tag{23}$$

where $P_s$ is the survival probability.

The sampled position will be:

$$x_n = x_{n-1} - \frac{\mu_{n-1} \ln \rho}{\beta \Sigma_t} \tag{24}$$

where: $\mu$ is the cosine of the scattering angle.

**EXERCISES**

1. Compare the results of the particle transport simulation to the case of a highly absorbing medium with an absorption probability of 90 percent. Is obtaining the exact analytical result using Monte Carlo easier or more difficult in this case?
2. Instead of predetermining the length of the generated particle chains at 30 terms, use the Russian Roulette procedure at a preset triggering weight to terminate the particle histories.